



Mechanizing Bisimulation Theorems for Relation-Changing Logics in Coq

Raul Fervari^{1,2(✉)}, Francisco Trucco¹, and Beta Ziliani^{1,2}

¹ FaMAF, Universidad Nacional de Córdoba, Córdoba, Argentina
fervari@famaf.unc.edu.ar, franciscostrucco@gmail.com, beta@mpi-sws.org

² Consejo Nacional de Investigaciones Científicas y Técnicas (CONICET),
Buenos Aires, Argentina

Abstract. Over the last years, the study of logics that can modify a model while evaluating a formula has gained in interest. Motivated by many examples in practice such as hybrid logics, separation logics and dynamic epistemic logics, the ability of updating a model has been investigated from a more general point of view. In this work, we formalize and verify in the proof assistant Coq, the bisimulation theorems for a particular family of dynamic logics that can change the accessibility relation of a model. The benefits of this formalization are twofold. First, our results apply for a wide variety of dynamic logics. Second, we argue that this is the first step towards the development of a modal logic library in Coq, which allows us to mechanize many relevant results in modal logics.

1 Introduction

Historically, *modal logic* [10, 11] has been understood as a logic to reason about different modes of truth. Under this perspective, it can be seen as an extension of propositional logic with *modalities*, that in certain contexts may have some particular interpretations, such as necessity, knowledge, belief, temporality, or obligation, just to name a few. Nowadays, *modal logics* is a term defining a family of logics to reason about relational structures, i.e., to reason about graphs. This is a consequence of the insights provided by the most common semantics for modal logics given in terms of the so-called Kripke structures [26]. In the wide spectrum of existing modal logics, one family has gained in interest over the last years: *dynamic modal logics*, i.e., logics that are able to update the model while evaluating the truth of a formula. Some well-known examples of this family are dynamic epistemic logics [37], separation logics [30, 31], and hybrid logics [6].

The aforementioned examples of dynamic logics are specific instances designed with a particular goal in mind. Over the last years, there has been an increasing interest in understanding the behaviour of dynamic logics from a more general point of view (see e.g. [3, 27, 34]). Such perspective allows us to investigate the properties of abstract operators that are the building blocks used to construct concrete modalities, and to obtain a general perspective of the impact of including such kind of operators.

Some examples of results about operators that can change the accessibility relation of a model are collected in [18]. In [1, 2] various abstract operations are presented, in particular, modalities to delete, add and swap-around an edge (both locally at an evaluation point and globally in the whole model). These logics are called *relation-changing logics*. In such works, particular notions of bisimulation for each operation are defined. A more general approach is taken in [3], where the notion of ‘updating a relation’ is generalized, and some results can be proved for all the logics encompassed in this framework. Also, the complexity of model checking is studied, whereas satisfiability is investigated in [4] (see also [8, 27]). All these results require proofs that are tedious and of high complexity, so it would be interesting to benefit of the use of computational tools in order to guide or verify (parts of) the proofs.

The *Coq proof assistant* [9] is an interactive tool that helps us to perform complex mathematical proofs. It provides a formal language to formalize mathematical definitions, algorithms, theorems and their proofs. One of the main advantages of the Coq assistant, is that it allows us to build mathematical proofs constructively. The underlying logic in Coq is an intuitionistic logic with dependent types, known as *the calculus of inductive constructions (CIC)*. Thanks to the Curry-Howard correspondence, propositions are interpreted as types and proofs are interpreted as programs with the type of the corresponding proposition. Thus, we can say that Coq is essentially a type verifier (see e.g. [25]).

In the last years, several mathematical problems have been solved with the help of interactive tools like Coq; problems whose pen-and-paper proofs were put to doubt due to their complexity. For instance, in [22] a problem from graph theory known as *the four colour problem* was solved with the assistance of Coq. More recently, in [23] the *Kepler conjecture*, an open problem from combinatorial geometry, was proved by using a combination of the proof assistants HOL light [33] and Isabelle [29].

Motivated by examples as those in the last paragraph, we aim to develop a library to formalize and verify formal proofs in modal logic. In particular, we extend the formalization provided in [15] in order to model a particular family of dynamic logics called *relation-changing modal logics*. We introduce a family of dynamic operators, parameterized by a *model update function* that given a relational model, it returns a modified relational model. This follows the definitions introduced in [3, 18]. Then we formalize a bisimulation notion which is agnostic with respect to the model update function, and mechanize the proof of the theorem of *invariance under bisimulation*. This theorem establishes that given two models that are related by a bisimulation, they satisfy the same formulas of the corresponding language. We consider this is the first step towards the development of a library for the mechanization of proofs for a wide variety of modal and dynamic logics.

Structure. In Sect. 2 we introduce the syntax and semantics of relation-changing modal logics, in which we have modalities parameterized by a model update function. In Sect. 3 we introduce the notion of bisimulation, and enunciate the invariance theorem. Then we focus on our main contribution: in

Sect. 4 we present the formalization of these results in Coq. We conclude in Sect. 5 with some final remarks and future lines of research.

2 Basic Definitions

The syntax of the family of dynamic modal logics we call herein *relation-changing modal logics*, is a straightforward extension of the propositional logic. Let us introduce their syntax and semantics.

Definition 1 (Syntax). *Let PROP be a countable, infinite set of propositional symbols. The set FORM of formulas over PROP is defined as:*

$$\text{FORM} ::= \perp \mid p \mid \varphi \rightarrow \psi \mid \blacklozenge_i \varphi,$$

where $p \in \text{PROP}$, $\blacklozenge_i \in \text{DYN}$ a set of dynamic operators, and $\varphi, \psi \in \text{FORM}$. Other Boolean operators are defined as usual. $\blacksquare_i \varphi$ is defined as $\neg \blacklozenge_i \neg \varphi$.

For $S \subseteq \text{DYN}$ a set of dynamic operators, we call $\mathcal{L}(S)$ the extension of the propositional language allowing also the operators in S . If S is a singleton set $S = \{\blacklozenge\}$, we write $\mathcal{L}(\blacklozenge)$ instead of $\mathcal{L}(\{\blacklozenge\})$.

Semantically, formulas of $\mathcal{L}(S)$ are evaluated in standard relational models.

Definition 2 (Models). *A model \mathcal{M} is a triple $\mathcal{M} = \langle W, R, V \rangle$, where W is the domain, a non-empty set whose elements are called points or states; $R \subseteq W \times W$ is the accessibility relation; and $V : \text{PROP} \mapsto 2^W$ is the valuation.*

Let w be a state in \mathcal{M} , the pair (\mathcal{M}, w) is called a pointed model; we usually drop parentheses and call \mathcal{M}, w a pointed model.

In this article, we restrict ourselves to models with only one accessibility relation (i.e., the underlying modal language has only one modal operator). A generalization to models with multiple accessibility relations is possible, but leads to further choices concerning the definition of the dynamic operators (e.g., which relation is affected by a given dynamic operator). Also, we only consider changes on the accessibility relation, but changes in the valuation would be easily incorporated in this framework.

Definition 3 (Model update functions). *Given a domain W , a model update function for W is a function $f_W : W \times 2^{W^2} \rightarrow 2^{W \times 2^{W^2}}$, that takes a state in W and a binary relation over W and returns a set of possible updates to the state of evaluation and accessibility relation.*

Let \mathcal{C} be a class of models, a family of model update functions f is a class of model update functions, one for each domain of a model in \mathcal{C} :

$$f = \{f_W \mid \langle W, R, V \rangle \in \mathcal{C}\}.$$

\mathcal{C} is closed under a family of model update functions f if whenever $\mathcal{M} = \langle W, R, V \rangle \in \mathcal{C}$, then $\{\langle W, R', V \rangle \mid f_W \in f, w \in W, (v, R') \in f_W(w, R)\} \subseteq \mathcal{C}$.

Clearly, the class of all pointed models is closed under any family of model update functions. In the rest of the article we only discuss the class of all models.

Notice, in the definition above, that a model update function is defined relative to a domain. We specifically require that all models with the same domain have the same model update function. This constraint limits the number of operators that can be captured in the framework, but at the same time leads to operators with a more uniform behavior. We will discuss this issue further after we introduce the formal semantics of the relation-changing operators below.

We now introduce the semantics for the general case.

Definition 4 (Semantics). *Let \mathcal{C} be a class of models, $\mathcal{M} = \langle W, R, V \rangle$ be a model in \mathcal{C} , $w \in W$ a state, f a family of model update functions for \mathcal{C} and \Diamond_f its associated dynamic modality. Let φ be a formula in $\mathcal{L}(\Diamond_f)$. We say that \mathcal{M}, w satisfies φ , and write $\mathcal{M}, w \models \varphi$, when*

$$\begin{aligned} \mathcal{M}, w &\models \perp && \text{never} \\ \mathcal{M}, w &\models p && \text{iff } w \in V(p) \\ \mathcal{M}, w &\models \varphi \rightarrow \psi && \text{iff } \mathcal{M}, w \not\models \varphi \text{ or } \mathcal{M}, w \models \psi \\ \mathcal{M}, w &\models \Diamond_f \varphi && \text{iff for some } (v, R') \in f_W(w, R), \langle W, R', V \rangle, v \models \varphi. \end{aligned}$$

The definition extends to languages with many modal dynamic operators in the obvious manner. A formula φ is satisfiable if for some pointed model \mathcal{M}, w we have $\mathcal{M}, w \models \varphi$. We write $\mathcal{M}, w \equiv_{\mathcal{L}} \mathcal{N}, v$ when both models satisfy the same \mathcal{L} -formulas, i.e., for all $\varphi \in \mathcal{L}$, $\mathcal{M}, w \models \varphi$ if and only if $\mathcal{N}, v \models \varphi$.

Notice, in the semantic definition, how the relation-changing modal operator \Diamond_f potentially changes both the state of evaluation and the accessibility relation. On the other hand, the domain remains the same, and hence all \Diamond_f operators in a formula are evaluated using the same model update function.

Examples of Relation-Changing Logics. First, notice that the classical modal diamond \Diamond [10,11] is one particular instance of a dynamic operator, in which the accessibility relation remains unchanged and the evaluation state is changed by some successor via R . To simplify notation we use wv as a shorthand for $\{(w, v)\}$ or (w, v) ; context will always disambiguate the intended use. Let W a domain and $R \subseteq W^2$, the model update function associated to \Diamond is defined as

$$f_W^\Diamond(w, R) = \{(v, R) \mid wv \in R\}.$$

Consider now the model update functions from [3]. Given a binary relation R , let us introduce some notation:

$$R_{wv}^- = R \setminus wv \quad R_{wv}^+ = R \cup wv \quad R_{wv}^* = (R \setminus vw) \cup wv.$$

Define the following six model update functions, which give rise to natural dynamic operators: Van Benthem's sabotage operator \Diamond_{gsb} [34], and a local version \Diamond_{sb} that deletes an existing edge between the current state of evaluation

and a successor state; a “bridge” operator $\blacklozenge_{\text{br}}$ that adds an edge between two previously unconnected states, and a local version $\blacklozenge_{\text{br}}$ that links the current state of evaluation and an inaccessible state; and the global and local versions ($\blacklozenge_{\text{gsw}}$ and $\blacklozenge_{\text{sw}}$, respectively) of an operation that swaps around edges of the model.

$$\begin{aligned} f_W^{\text{sb}}(w, R) &= \{(v, R_{wv}^-) \mid wv \in R\} & f_W^{\text{gsw}}(w, R) &= \{(w, R_{uv}^-) \mid uv \in R\} \\ f_W^{\text{br}}(w, R) &= \{(v, R_{wv}^+) \mid wv \notin R\} & f_W^{\text{gbr}}(w, R) &= \{(w, R_{uv}^+) \mid uv \notin R\} \\ f_W^{\text{sw}}(w, R) &= \{(v, R_{vw}^*) \mid wv \in R\} & f_W^{\text{gsw}}(w, R) &= \{(w, R_{vu}^*) \mid uv \in R\}. \end{aligned}$$

It is easy to show that the basic modal logic $\mathcal{L}(\blacklozenge)$ enriched with other relation-changing modalities gains in expressivity. For example, the local sabotage operator $\blacklozenge_{\text{sb}}$ and the local swap operator $\blacklozenge_{\text{sw}}$ are logically stronger than the diamond operator when restricted to non-dynamic predicates, as the formulas $\blacklozenge_{\text{sb}}p \rightarrow \blacklozenge p$ and $\blacklozenge_{\text{sw}}p \rightarrow \blacklozenge p$ are valid. These operators are very expressive as they can force non-tree models (see e.g. [3, 18]). For example, the formula $\blacksquare_{\text{sb}}\Box\perp$ means that any local sabotage leads to a dead-end, hence the formula $\blacklozenge\blacklozenge\top \wedge \blacksquare_{\text{sb}}\Box\perp$ can only be true at a reflexive state, a property that cannot be expressed in $\mathcal{L}(\blacklozenge)$.

3 Bisimulations

In modal model theory, the notion of bisimulation is a crucial tool. Typically, a bisimulation is a binary relation linking elements of the domains that have the same atomic information, and preserving the relational structure of the model. Because we need to keep track of the changes on the accessibility relation that the dynamic operators may introduce, bisimulations are defined as relations that link pairs of a state together with the current accessibility relation [3, 18]. Notice that the notion we introduce is parameterized with a model update function, making the results general for the relation-changing logics from Sect. 2.

Definition 5 (Bisimulations). *Let $\mathcal{M} = \langle W, R, V \rangle$, $\mathcal{M}' = \langle W', R', V' \rangle$ be two models, and f a family of model update functions. A non empty relation $Z \subseteq (W \times 2^{W^2}) \times (W' \times 2^{W'^2})$ is an $\mathcal{L}(\blacklozenge_f)$ -bisimulation if it satisfies the following conditions. If $(w, S)Z(w', S')$ then*

- (atomic harmony) *for all $p \in \text{PROP}$, $w \in V(p)$ iff $w' \in V'(p)$;*
- (f -zig) *if $(v, T) \in f_W(w, S)$, there is $(v', T') \in f_{W'}(w', S')$ s.t. $(v, T)Z(v', T')$;*
- (f -zag) *if $(v', T') \in f_{W'}(w', S')$, there is $(v, T) \in f_W(w, S)$ s.t. $(v, T)Z(v', T')$.*

Given two pointed models \mathcal{M}, w and \mathcal{M}', w' they are $\mathcal{L}(\blacklozenge_f)$ -bisimilar (notation, $\mathcal{M}, w \rightleftharpoons_{\mathcal{L}(\blacklozenge_f)} \mathcal{M}', w'$) if there is an $\mathcal{L}(\blacklozenge_f)$ -bisimulation Z such that $(w, R)Z(w', R')$ where R and R' are the respective relations of \mathcal{M} and \mathcal{M}' .

Summing up, the bisimulation notion for each logic $\mathcal{L}(\blacklozenge_f)$ includes (atomic harmony) and the particular conditions for the model update function f . For instance, according to the above definition, the (zig) and (zag) conditions for the basic modal logic $\mathcal{L}(\blacklozenge)$ are defined as:

(**zig**) if $(w, v) \in S$, there is $v' \in W'$ s.t. $(w', v') \in S'$ and $(v, S)Z(v', S')$;
 (**zag**) if $(w', v') \in S'$, there is $v \in W$ s.t. $(w, v) \in S$ and $(v, S)Z(v', S')$.

On the other hand, instantiating f with f^{sb} we get the following conditions:

(f^{sb} -**zig**) If $(w, v) \in S$, there is $v' \in W'$ s.t. $(w', v') \in S'$ and $(v, S_{vv}^-)Z(v', S_{w'v'}^-)$;
 (f^{sb} -**zag**) If $(w', v') \in S'$, there is $v \in W$ s.t. $(w, v) \in S$ and $(v, S_{vv}^-)Z(v', S_{w'v'}^-)$.

In the same way, we can instantiate f with any of the model update functions mentioned in Sect. 2.

Theorem 1 (Invariance). *Let f be a family of model update functions, then $\mathcal{M}, w \Leftrightarrow_{\mathcal{L}(\Diamond_f)} \mathcal{M}', w'$ implies $\mathcal{M}, w \equiv_{\mathcal{L}(\Diamond_f)} \mathcal{M}', w'$.*

Proof. Let $\mathcal{M} = \langle W, R, V \rangle$ and $\mathcal{M}' = \langle W', R', V' \rangle$, such that $\mathcal{M}, w \Leftrightarrow_{\mathcal{L}(\Diamond_f)} \mathcal{M}', w'$. Then there exists Z such that $(w, R)Z(w', R')$.

We prove the theorem by structural induction. In fact, we prove a more general result. Let $S \subseteq W^2$, $S' \subseteq W'^2$ such that $(w, S)Z(w', S')$, we will show that $\langle W, S, V \rangle, w \equiv_{\mathcal{L}(\Diamond_f)} \langle W', S', V' \rangle, w'$. The base cases hold by (atomic harmony), and the \rightarrow case is trivial.

[$\Diamond_f \varphi$ case:] Suppose $\langle W, S, V \rangle, w \models \Diamond_f \varphi$. Then there is $(v, T) \in f_W(w, S)$ s.t. $\langle W, T, V \rangle, v \models \varphi$. Because Z is a bisimulation, by (f -zig) we have $(v', T') \in f_{W'}(w', S')$ s.t. $(v, T)Z(v', T')$. By inductive hypothesis, $\langle W', T', V' \rangle, v' \models \varphi$ and by definition, $\langle W', S', V' \rangle, w' \models \Diamond_f \varphi$. For the other direction use (f -zag).

Therefore, since $(w, R)Z(w', R')$, we get $\mathcal{M}, w \equiv_{\mathcal{L}(\Diamond_f)} \mathcal{M}', w'$. \square

Clearly the result holds when we extend \mathcal{L} with any set of relation-changing modal operators. It suffices to require that the bisimulation comply with the various (f -zig) and (f -zag) conditions corresponding to all operators. In the next section we will reproduce the proof of Theorem 1 in Coq.

4 Formalization in Coq

4.1 The Coq Proof Assistant in a Nutshell

A typical proof in Coq looks like the following:

Lemma and **intro**: **forall** (**A B** : **Prop**), **A** \rightarrow **B** \rightarrow **A** \wedge **B**.

Proof.

intros **A B** **HA HB**. split.

– apply **HA**.

– apply **HB**.

Qed.

This simple proof states that if you are given a proof of proposition **A** and another proof of proposition **B**, then you have a proof for their conjunction. In order to be able to state the lemma and prove it, Coq presents three different domain specific languages: Gallina, The Vernacular and the tactics language Ltac. Gallina is Coq's mathematical higher-level language and program specification language.

Seen as a programming language, Gallina is a dependently-typed functional language, while seen as a logical system, Gallina is a higher-order type theory. In the example, the lemma's statement (what follows the `:`) is written in this language.

The Vernacular allows the definition of functions or predicates, the statement of mathematical theorems and software specifications, the machine checking of proofs and the extraction of certified programs to different languages. In the example we use the following Vernacular *commands*: `Lemma` indicates the desire to state a theorem; `Proof` starts the proof; and `Qed` signals that the proof is completed, and therefore must be checked for errors and stored in the database of known facts if everything is correct. The reason for this check is to guarantee that the proof is indeed complete and that the tactics used to write the proof (see below) rightfully solved the problem.

Finally, the proof itself is written using the tactic language Ltac. In formal reasoning, a deduction rule is a link between a conclusion and a list of premises. There are two ways to understand a deduction rule. With *forward reasoning*, if we want to deduce the conclusion, we first try to deduce the list of premises and then use the deduction rule to prove the conclusion. With *backward reasoning* we go in the opposite direction: in order to prove the conclusion, we must prove the premises. Coq tactics are typically deduction rules that implement backward reasoning: when applied to a conclusion, usually called a *goal*, a tactic replaces this goal with the *subgoals* it generates, one for each premise of the rule. In the example, everything between `Proof` and `Qed` are tactic invocations. For instance, the tactic `split` replaces the goal $A \wedge B$ with two subgoals, one for proving A and another for proving B .

Not all tactics are as simple as `split`. For instance, the tactic `tauto` implements a decision procedure for intuitionistic propositional calculus, so it is appropriate to solve many trivial statements (actually, it is capable of solving the above lemma in just one tactic invocation). Ltac also allows the definition of complex user defined tactics and decision procedures. See e.g. [9] for a more complete presentation of Coq.

4.2 Formalizing Relation-Changing Logics in Coq

In the rest of this section we will present our formalization of relation-changing modal logics in Coq, and the mechanization of the proof of the general invariance theorem (Theorem 1). The source code can be found at <http://tinyurl.com/rcml-in-coq0>. As we will see in the formalization below, it is easy to match the mathematical definitions from Sects. 2 and 3 with their counterpart in Coq.

Syntax. In order to formalize the syntax of relation-changing modal logics, we first need to define the countable set of propositional symbols `PROP`. We can accomplish this by using an inductive type definition:

```
Inductive prop : Set := p : nat → prop.
```

This creates a new type called `prop` with a type constructor `p` that given a natural number `n` constructs an inhabitant of the type `prop`, namely `p n`. It is clear that `prop` correctly formalizes the countable infinite set of propositions `PROP`.

Before we give the definition of the syntax we need to assume that a set of dynamic operators actually exists. This assumption is necessary because the definition of $\mathcal{L}(S)$, with $S \subseteq \text{DYN}$, depends on the existence of a set of dynamic operators DYN .

Variable $\text{Dyn} : \text{Set}$.

Now we can give the definition of the set of formulae **FORM**, as in Definition 1:

```
Inductive form : Type :=
| Atom    : prop → form
| Bottom  : form
| If       : form → form → form
| DynDiam : Dyn → form → form.
```

Each line of this definition is interpreted as the members of the BNF from Definition 1. Other operators are defined as syntactic sugar. For example:

Definition $\text{DynBox } (d : \text{Dyn}) \text{ (phi : form) : form} := \text{Not } (\text{DynDiam } d \text{ (Not phi)})$.

Writing formulae with these constructors can be very tedious. Fortunately, Coq allows us to define a notation for them (we present the cases of dynamic operators; other operators are defined as expected):

Notation " $\langle o \rangle d \text{ phi}$ " := $(\text{DynDiam } d \text{ phi})$ (at level 65, right associativity).

Note that between parentheses we specify the precedence level and associativity.

Semantics. We now turn our attention to the formalization of the semantics of these logics. First, we need to formalize the concept of a relational model as those from Definition 2. In turn this requires to consider how powersets and relations are represented in Coq.

Let A be a set and \mathbf{A} its corresponding formalization in Coq. In order to formalize a subset S of the power set 2^A , we can view S as a function that for each element $a \in A$ determines whether a belongs to S or not. Naively, one could think that S can be modeled with a function from \mathbf{A} to `bool`. However, this is overly restrictive, as it will force us to make S decidable (Coq's functions are guaranteed to terminate). Thus, we use the constructive type **Prop** instead of `bool` and write $\mathbf{A} \rightarrow \text{Prop}$ to mean "a subset of the 2^A ".

Similarly, a binary relation R over A can be viewed as a function that given two elements $a, b \in A$ determines whether aRb or not. For this reason, we model R as $\mathbf{A} \rightarrow \mathbf{A} \rightarrow \text{Prop}$. Or, using Coq's standard library, simply **relation** \mathbf{A} .

Now we are ready to introduce the formalization of the models of our logic. We can think a relational model as a triple consisting of a set \mathbf{W} , a binary relation \mathbf{R} defined over \mathbf{W} , and a valuation function that for each element in \mathbf{W} and each propositional symbol with type **prop**, decides whether that propositional symbol is valid or not in that element of \mathbf{W} . In Coq we write the types of \mathbf{W} , \mathbf{R} and \mathbf{V} as:

```
W : Set
R : relation W
V : W → prop → Prop
```


The type of the valuation function deserves an explanation. Above we define the valuation function as a function $V : \text{PROP} \rightarrow 2^W$. Given a propositional symbol $p \in \text{PROP}$, an element $w \in W$ and their respective formalizations $\mathbf{p} : \text{prop}$ and $\mathbf{w} : \mathbf{W}$, we formalize the expression $w \in V(p)$ as $\mathbf{V} \ \mathbf{w} \ \mathbf{p}$.

Before we can give a definition of satisfiability we must give a formalization for the type of all model update functions. Remember that a model update function for a given domain W is a function $f_W : W \times 2^{W^2} \rightarrow 2^{W \times 2^{W^2}}$ that for each state of W and for each binary relation over W , associates a set of possible updates to the evaluation state and the accessibility relation. In Coq, we define the type of a model update function (`muf`) as:

Definition `point` ($\mathbf{W} : \text{Set}$) : `Type` := ($\mathbf{W} * \text{relation } \mathbf{W}$).

Definition `muf` : `Type` := `forall` ($\mathbf{W} : \text{Set}$),
(`point` \mathbf{W}) \rightarrow (`point` $\mathbf{W} \rightarrow \text{Prop}$).

As with the mathematical definition f_W , a `muf` depends on the set W , and that is why we start with `forall` ($\mathbf{W} : \text{Set}$), For readability, we define a type `point W` to denote ($\mathbf{W} * \text{relation } \mathbf{W}$), the Coq equivalent of $W \times 2^{W^2}$. Then, we define the `muf` as (`point` \mathbf{W}) \rightarrow (`point` $\mathbf{W} \rightarrow \text{Prop}$). Note that the parentheses are just for readability: the function type \rightarrow is right-associative.

Given that we have defined the notion of model update function and both the syntax and the models of our logic, we can now define the notion of satisfiability. It must be clear at this point that in order to define it, we need a function that assigns to each dynamic operator a model update function. We can simply assume that such function exists.

Variable $\mathbf{F} : \text{Dyn} \rightarrow \text{muf}$.

Fixpoint `satisfies` ($\mathbf{W} : \text{Set}$) ($\mathbf{R} : \text{relation } \mathbf{W}$) ($\mathbf{V} : \mathbf{W} \rightarrow \text{prop} \rightarrow \text{Prop}$)
($\mathbf{w} : \mathbf{W}$) ($\mathbf{phi} : \text{form}$) : `Prop` :=
`match` \mathbf{phi} `with`
| `Atom` $\mathbf{a} \Rightarrow \mathbf{V} \ \mathbf{w} \ \mathbf{a}$
| `Bottom` $\Rightarrow \text{False}$
| `If` $\mathbf{phi1} \ \mathbf{phi2} \Rightarrow (\text{satisfies } \mathbf{W} \ \mathbf{R} \ \mathbf{V} \ \mathbf{w} \ \mathbf{phi1}) \rightarrow (\text{satisfies } \mathbf{W} \ \mathbf{R} \ \mathbf{V} \ \mathbf{w} \ \mathbf{phi2})$
| `DynDiam` $\mathbf{d} \ \mathbf{phi} \Rightarrow$
 `let` $\mathbf{fw} := \mathbf{F} \ \mathbf{d} \ \mathbf{W}$ `in`
 `exists` ($\mathbf{v} : \mathbf{W}$) ($\mathbf{R}' : \text{relation } \mathbf{W}$), $\mathbf{fw} \ (\mathbf{w}, \ \mathbf{R}) \ (\mathbf{v}, \ \mathbf{R}') \ \wedge \text{satisfies } \mathbf{W} \ \mathbf{R}' \ \mathbf{V} \ \mathbf{v} \ \mathbf{phi}$
`end`.

Note how each case has one-to-one correspondence with Definition 4. For readability, we define the following notation:

Notation `"# W , R , V >> w |= phi"` := (`satisfies` $\mathbf{W} \ \mathbf{R} \ \mathbf{V} \ \mathbf{w} \ \mathbf{phi}$) (at level 30).

Properties. With all of these definitions we can now formalize the notions of modal equivalence and bisimulation. For this part, we assume we work under the following assumptions:

Variables $\mathbf{W} \ \mathbf{W}' : \text{Set}$.

Variables ($\mathbf{V} : \mathbf{W} \rightarrow \text{prop} \rightarrow \text{Prop}$) ($\mathbf{V}' : \mathbf{W}' \rightarrow \text{prop} \rightarrow \text{Prop}$).

The notion of modal equivalence can be formalized unsurprisingly as:

```
Definition equivalent_at_points '(w, R) '(w', R') :=
  forall (phi:form), (# W , R , V >> w |= phi) ↔ (# W' , R' , V' >> w' |= phi).
```

(The notation $'(a, b)$ just serves to say the definition takes a pair with components a and b .)

Before formalizing the notion of bisimulation for these logics, let us define the type of relations between models. As introduced in Definition 5, the type of relations defining bisimulations relates pairs of points and binary relations over the domains of the models:

```
Definition model_to_model_relation : Type :=
  (point W) → (point W') → Prop.
```

To formalize the notion of bisimulation, we first define each condition separately and then use them as functions in the definition of bisimulation (`f_zag` clause is analogous). To state these conditions, we work under the assumption that we have a relation between models Z :

```
Variable Z : model_to_model_relation.
```

```
Definition atomic_harmony : Prop :=
  forall w S w' S', Z (w, S) (w', S') → V w = V' w'.
```

```
Definition f_zig (f : muf) : Prop :=
  forall w S w' S' v T, Z (w, S) (w', S') →
    f W (w, S) (v, T) →
      (exists (v' : W') T', f W' (w', S') (v', T') ∧ Z (v, T) (v', T')).
```

Each condition shares the precondition of bisimulation (see Definition 5), namely that there is a relation Z between the models, and that the condition holds for every states w and w' and relations S and S' such that they are related according to Z .

Finally, the notion of bisimulation is defined as follows:

```
Definition bisimulation : Prop := atomic_harmony ∧
  (forall d : Dyn, (f_zig (F d))) ∧ (forall d : Dyn, (f_zag (F d))).
```

```
Definition bis_at_points (p: point W) (p': point W') : Prop :=
  bisimulation ∧ Z p p'.
```

Now we are ready to formally state the Theorem of Invariance under Bisimulation (Theorem 1):

```
Theorem InvarianceUnderBisimulation :
  forall (p: point W) (p': point W'),
    bis_at_points p p' → equivalent_at_points p p'.
```

Proof. The proof follows by structural induction on the formula ϕ . In order to get the right induction principle, we need to first massage the goal and the list of hypotheses a bit. The first lines of the proof are the following (we use Coq's `(* comments *)` to explain each line).

```

intros [w S] [w' S'].          (* name each component of the points *)
unfold bis_at_points.          (* unfold definitions *)
unfold equivalent_at_points.
unfold bisimulation.

```

At this point, the goal looks like

```

(atomic_harmony ∧
 (forall d : Dyn, f_zig (F d)) ∧ (forall d : Dyn, f_zag (F d))) ∧
Z (w, S) (w', S') →
forall phi : form, # W, S, V >> w |= phi ↔ # W', S', V' >> w' |= phi

```

We split the different components of the first hypothesis, naming each of them again with the `intros` tactic. Then we introduce the formula `phi`.

```

intros [[HAtomicHarmony [HFZig HFZag]] HZwSw'S'].
intros phi.

```

Now our hypotheses looks like follows (the ... omits the trivial ones):

```

...
HAtomicHarmony : atomic_harmony
HFZig : forall d : Dyn, f_zig (F d)
HFZag : forall d : Dyn, f_zag (F d)
HZwSw'S' : Z (w, S) (w', S')
phi : form

```

With the goal being $\# W, S, V \gg w \models \text{phi} \leftrightarrow \# W', S', V' \gg w' \models \text{phi}$. We are almost ready to use structural induction on `phi`, but first we need to strengthen our inductive hypothesis, so we can use it on any points (i.e., on any pairs composed by a state and a binary relation). We do this by generalizing the goal:

```

generalize dependent S'. generalize dependent S.
generalize dependent w'. generalize dependent w.

```

The goal now looks like

```

forall (w : W) (w' : W') (S : relation W) (S' : relation W'),
Z (w, S) (w', S') →
# W, S, V >> w |= phi ↔ # W', S', V' >> w' |= phi

```

Now, we are ready to perform structural induction on `phi`:

```

induction phi as [p | | phi IHphi psi IHpsi | d phi IH];

```

The syntax above tells Coq how it should name the different hypothesis in the different cases. Notice that we usually end the tactics with a dot (.) but this time we ended the tactic `induction` with a semicolon (;). This tells Coq that the next tactic has to be applied to all cases in the induction. In particular we unfold the definition of `satisfies` and introduce all required variables and hypothesis.

```

simpl;          (* This tactic unfolds definitions *)
intros w w' S S' HZwSw'S'.

```

The atomic case is $V \ w \ p \leftrightarrow V' \ w' \ p$, which is solved with a simple rewrite using Atomic Harmony and a call to the tactic `tauto` mentioned in the previous section.

```
rewrite (HAtomicHarmony w S w' S' HZwSw'S').
tauto.      (* Solves the goal "V' w' p ↔ V' w' p" *)
```

For the bottom case we simply use `tauto`. Now for the if case, we split the proof into two separate directions. First we prove the left-to-right direction and then the right-to-left. This is performed using the `split` tactic. Luckily, we do not need to think about the two directions separately, since the same tactics work to prove both directions. That is why we end all tactics with a semicolon. Both directions are proved simply by assuming the two antecedents and then applying the inductive hypothesis of `psi` and `phi` and these two antecedents to the current subgoal.

```
split;
intros HIf Hsat;
apply (IHpsi w w' S S' HZwSw'S');
apply HIf;
apply (IHphi w w' S S' HZwSw'S');
apply Hsat.
```

The proof of the dynamic operator follows closely the same reasoning as the one present in Sect. 3. First, like with the if case, we use the tactic `split` to consider both directions, but this time we prove them separately. We will explain the left-to-right direction only, as the other direction is analogous. After `split` and `simpl` Coq tells us that we need to prove the following:

```
(exists (v : W) (R' : W → W → Prop),
  F d W (w, S) (v, R') ∧ # W, R', V >> v |= phi) →
exists (v : W') (R' : W' → W' → Prop),
  F d W' (w', S') (v, R') ∧ # W', R', V' >> v |= phi
```

First we give a name to the existentially quantified values of the antecedent, together with its properties:

```
intros [v [T [HfWwSvT HsatTv]]].
```

We get that $v : W$, that $T : \text{relation } W$ and that they also satisfy two properties that we will name `HfWwSvT` and `HsatTv`:

```
HfWwSvT : F d W (w, S) (v, T)      (* (v,T) is a successor via F d of (w,S) *)
HsatTv : # W, T, V >> v |= phi    (* the updated model satisfies phi *)
```

At this point we can apply the `HFZig` hypothesis in `HfWwSvT`:

```
apply (HFZig d w S w' S' v T HZwSw'S') in HfWwSvT
as [v' [T' [HfW'w'S'v'T' HZvTv'T']]].
```

This introduces the following new hypotheses:

```
v' : W'
T' : relation W'
HfW'w'S'v'T' : F d W' (w', S') (v', T')
HZvTv'T' : Z (v, T) (v', T')
HsatTv : # W, T, V >> v |= phi
```

Remember that our goal at this point is to prove that:

```
exists (v'' : W') (R' : relation W'),
  F d W' (w', S') (v'', R')  $\wedge$  # W', R', V' >> v'' |= phi
```

The existential quantifiers are removed by providing witnesses: v' and T' :

```
exists v'. exists T'.
```

Now we only need to prove that:

```
F d W' (w', S') (v', T')  $\wedge$  # W', T', V' >> v' |= phi
```

The left proposition of the conjunction is identical to one of our hypothesis (namely $HfW'w'S'v'T'$). The other proposition can be proved by applying the inductive hypothesis to the $HsatTv$ hypothesis. We can prove all this using the following tactics:

```
split.
* assumption.
* eapply IH. apply HZvTv'T'. assumption.
```

The proof is ended, so we issue the closing command `Qed`.

5 Final Remarks

In this work we formalized in the interactive proof assistant Coq, the syntax, semantics and a notion of bisimulation for a family of dynamic logics called *relation-changing logics*. These logics contain modalities that update the accessibility relation of the model while evaluating a formula. One particularity of our formalization is that, following the definition from [3], dynamic modalities are parameterized by a *model update function*, i.e., a function that given a pointed model, returns an updated pointed model. Thus, the definitions in Coq are simple, but powerful enough to encompass a whole family of logics. With these definitions at hand, we recreated the proof of the *invariance under bisimulation theorem*: if two models are bisimilar for a determined logic, then they satisfy the same formulas. Again, given the generality of our framework, we only need to prove one theorem, which holds for any instance of model update function (e.g., sabotage logics [32,34], swap logic [2], and others such as the program-based relation-changing operations from [19–21,35], which are as expressive as *propositional dynamic logic* [24] and whose bisimulation notion is the same as for the basic modal logic [10]). Moreover, the results we introduced can be straightforwardly extended to model update functions that also update the valuation of the model (see e.g. [5,7,36]).

There exist in the literature other works exploring the mechanization of proofs for modal and non-classical logics. In [15], a formalization of the basic modal logic $\mathcal{L}(\Diamond)$ is presented, in which we based our formalization. In addition, the author formalizes the extensions of the basic modal logic $S5$ and $S5^n$ [10], together with a natural deduction system. Then, the system is used to solve some logical puzzles. Regarding non-classical logics, in [39] the authors present a formalization in Coq of *linear logic*, together with the mechanization of some

theorems for such logic, such as a proof of cut-elimination. A formalization in Coq of a Sahlqvist’s global correspondence theorem for the very simple Sahlqvist class is presented in [13]. From such formalization, it is possible to extract a verified Haskell program that computes correspondents of simple Sahlqvist formulas. Another approach has been taken in [38], in which a formalization and verification in Lean [14] of tableaux methods for modal logics is presented. Finally, recent works present a proof language for *differential dynamic logic* [12] for applications in cyber-physical systems, in the theorem prover KeYmaera X [28].

There are several interesting directions that we would like to explore in the future. The next step will be the mechanization of more complex results for relation-changing logics, such as the *Hennessy-Milner theorem* (i.e., the other direction of the proof presented here), and the correctness of the encodings into first-order and second-order logic [3]. Some of these results can be proved for particular instances of model updates functions, while others can be proved for the general case. Moreover, we would like to use our framework for mechanizing proofs for more concrete instances of this family, such as *dynamic epistemic logics* [37] and *modal separation logics* [16, 17].

We consider that this work represents a first step towards a more serious use of an interactive proof assistant for proofs in modal logic. One of the main problems in the aforementioned works, is that all the formalizations are too heterogeneous, and it makes difficult to reuse previous results. Our main goal is the development of a modal logic library that allows us the mechanization in a uniform way of a wide variety of results in modal and dynamic logics.

Acknowledgements. This work was partially supported by ANPCyT-PICTs-2017-1130 and 2016-0215, MinCyT Córdoba, SeCyT-UNC, and the Laboratoire International Associé INFINIS.

References

1. Areces, C., Fervari, R., Hoffmann, G.: Moving arrows and four model checking results. In: Ong, L., de Queiroz, R. (eds.) WoLLIC 2012. LNCS, vol. 7456, pp. 142–153. Springer, Heidelberg (2012). https://doi.org/10.1007/978-3-642-32621-9_11
2. Areces, C., Fervari, R., Hoffmann, G.: Swap logic. *Log. J. IGPL* **22**(2), 309–332 (2014)
3. Areces, C., Fervari, R., Hoffmann, G.: Relation-changing modal operators. *Log. J. IGPL* **23**(4), 601–627 (2015)
4. Areces, C., Fervari, R., Hoffmann, G., Martel, M.: Satisfiability for relation-changing logics. *J. Log. Comput.* **28**(7), 1443–1470 (2018)
5. Areces, C., Figueira, D., Figueira, S., Mera, S.: The expressive power of memory logics. *Rev. Symb. Log.* **4**(2), 290–318 (2011)
6. Areces, C., ten Cate, B.: Hybrid logics. In: Blackburn, P., Wolter, F., van Benthem, J. (eds.) *Handbook of Modal Logic*, pp. 821–868. Elsevier, Amsterdam (2007)
7. Aucher, G., Balbiani, P., Fariñas del Cerro, L., Herzig, A.: Global and local graph modifiers. *ENTCS* **231**, 293–307 (2009)

8. Aucher, G., van Benthem, J., Grossi, D.: Modal logics of sabotage revisited. *JLC* **28**(2), 269–303 (2018)
9. Bertot, Y., Castéran, P.: Interactive Theorem Proving and Program Development: Coq’Art The Calculus of Inductive Constructions, 1st edn. Springer, Heidelberg (2010)
10. Blackburn, P., de Rijke, M., Venema, Y.: Modal Logic. Cambridge Tracts in Theoretical Computer Science. Cambridge University Press, Cambridge (2001)
11. Blackburn, P., van Benthem, J.: Modal logic: a semantic perspective. In: Handbook of Modal Logic, pp. 1–84. Elsevier (2007)
12. Bohrer, B., Platzer, A.: Toward structured proofs for dynamic logics. *CoRR*, abs/1908.05535 (2019)
13. D’Abbrera, C., Goré, R.: Verified synthesis of (very simple) Sahlqvist correspondents via Coq. In: AiML 2018, Short Presentations, pp. 26–30. College Publications (2018)
14. de Moura, L., Kong, S., Avigad, J., van Doorn, F., von Raumer, J.: The lean theorem prover (system description). In: Felty, A.P., Middeldorp, A. (eds.) CADE 2015. LNCS (LNAI), vol. 9195, pp. 378–388. Springer, Cham (2015). https://doi.org/10.1007/978-3-319-21401-6_26
15. de Wind, P.: Modal Logic in Coq. Vrije Universiteit, Amsterdam (2001)
16. Demri, S., Fervari, R.: On the complexity of modal separation logics. In: AiML 2018, pp. 179–198. College Publications (2018)
17. Demri, S., Fervari, R., Mansutti, A.: Axiomatizing logics with separating conjunction and modalities. In: Calimeri, F., Leone, N., Manna, M. (eds.) JELIA 2019. LNCS (LNAI), vol. 11468, pp. 692–708. Springer, Cham (2019). https://doi.org/10.1007/978-3-030-19570-0_45
18. Fervari, R.: Relation-Changing Modal Logics. Ph.D. thesis, Universidad Nacional de Córdoba, Argentina (2014)
19. Fervari, R., Velázquez-Quesada, F.R.: Dynamic epistemic logics of introspection. In: Madeira, A., Benevides, M. (eds.) DALI 2017. LNCS, vol. 10669, pp. 82–97. Springer, Cham (2018). https://doi.org/10.1007/978-3-319-73579-5_6
20. Fervari, R., Velázquez-Quesada, F.R.: Introspection as an action in relational models. *JLAMP* **108**, 1–23 (2019)
21. Girard, P., Seligman, J., Liu, F.: General dynamic dynamic logic. In: AiML 2012, pp. 239–260. College Publications (2012)
22. Gonthier, G.: Formal proof – the four-color theorem (2008)
23. Hales, T.C., et al.: A formal proof of the kepler conjecture. *Forum Math. Pi* **5**, e2 (2017)
24. Harel, D.: Dynamic Logic. Foundations of Computing. The MIT Press, Cambridge (2000)
25. Howard, W.A.: The formulae-as-types notion of construction. To HB Curry: Essays Combin. Log. Lambda Calc. Formalism **44**, 479–490 (1980)
26. Kripke, S.: Semantical analysis of modal logic I. Normal propositional calculi. *Z. für Math. Log. Grundlagen der Math.* **9**, 67–96 (1963)
27. Löding, C., Rohde, P.: Model checking and satisfiability for sabotage modal logic. In: Pandya, P.K., Radhakrishnan, J. (eds.) FSTTCS 2003. LNCS, vol. 2914, pp. 302–313. Springer, Heidelberg (2003). https://doi.org/10.1007/978-3-540-24597-1_26
28. Mitsch, S., Platzer, A.: The keymaera X proof IDE - concepts on usability in hybrid systems theorem proving. In: F-IDE@FM 2016. EPTCS, vol. 240, pp. 67–81 (2016)
29. Paulson, L.C. (ed.): Isabelle: A Generic Theorem Prover, vol. 828. Springer, Heidelberg (1994). <https://doi.org/10.1007/BFb0030541>

30. Pym, D., Spring, J., O'Hearn, P.W.: Why separation logic works. *Philos. Technol.* **32**(3), 483–516 (2019)
31. Reynolds, J.C.: Separation logic: a logic for shared mutable data structures. In: *LICS 2002*, pp. 55–74. IEEE (2002)
32. Rohde, P.: On games and logics over dynamically changing structures. Ph.D. thesis, RWTH Aachen (2006)
33. Slind, K., Norrish, M.: A brief overview of HOL4. In: Mohamed, O.A., Muñoz, C., Tahar, S. (eds.) *TPHOLs 2008*. LNCS, vol. 5170, pp. 28–32. Springer, Heidelberg (2008). https://doi.org/10.1007/978-3-540-71067-7_6
34. Benthem, J.: An essay on sabotage and obstruction. In: Hutter, D., Stephan, W. (eds.) *Mechanizing Mathematical Reasoning*. LNCS (LNAI), vol. 2605, pp. 268–276. Springer, Heidelberg (2005). https://doi.org/10.1007/978-3-540-32254-2_16
35. van Benthem, J., Liu, F.: Dynamic logic of preference upgrade. *J. Appl. Non-Class. Log.* **17**(2), 157–182 (2007)
36. van Ditmarsch, H., van der Hoek, W., Kooi, B.: Dynamic epistemic logic with assignment. In: *AAMAS 2005*, pp. 141–148. ACM (2005)
37. van Ditmarsch, H., van der Hoek, W., Kooi, B.: *Dynamic Epistemic Logic*. Synthese Library. Springer, Heidelberg (2007). <https://doi.org/10.1007/978-1-4020-5839-4>
38. Wu, M., Goré, R.: Verified decision procedures for modal logics. In: *ITP 2019*. LIPIcs, vol. 141, pp. 31:1–31:19. Schloss Dagstuhl - Leibniz-Zentrum für Informatik (2019)
39. Xavier, B., Olarte, C., Reis, G., Nigam, V.: Mechanizing focused linear logic in Coq. *ENTCS* **338**, 219–236 (2018)